Edward F. Puccinelli
NASA/Goddard Space Flight Center
Greenbelt, MD 20771

Ground Location of Satellite Scanner Data

Formulation and computer coding, which determine the ground location of an area scanned by satellite-borne sensors, are given.

Introduction

A common problem in analysis of data generated by satellite scanners is to determine the location of the intersect of the scanner's line of sight with the body about which it orbits.

Earth-based examples include determining the location of data sensed by LANDSAT's Multi-Spectral Scanner (MSS) and the Nimbus Electrically Scanning Microwave Radiometer (ESMR).

Although the problem is not inherently difficult to formulate and solve mathematically, different techniques have been developed by independent analysts and programmers. The varieties of approaches can (and do) lead to inconsistencies in results as well as in computer run times.

Abstract: *This paper presents simple and accurate mathematical formulation for determining the ground location of remote sensor data. The techniques used are based on elementary concepts of differential geometry and lead to the development of a relation that gives location as a function of surface ellipticity, satellite position, velocity, attitude, and scanner orientation. The formula lends itself to simply computer coding and will hopefully lead to a standardization of the various techniques which have been developed to solve this problem.*

This paper will present a simple, accurate formulation of the problem which can be efficiently coded on a computer. The formulation will also be generalized so as not to be dependent upon any one particular satellite or experiment. Hopefully, this effort will lead to a standardization of the techniques (and results) used to solve the problem, as well as to elimination of the need to redevelop the mathematical formulas.

Overview of Problem

The general approach taken can be most simply described as follows: Given the spacecraft position, $\vec{s}$, the spacecraft velocity, $\vec{v}$, and the description of the surface B about which the satellite orbits, measured relative to a basic coordinate system centered within a body (e.g., the earth); and yaw, y, pitch, p, roll, r, and the pointing direction of the scanner, $\vec{w}$, measured relative to the spacecraft axes; then it is possible to determine a function of these variables which gives the point of intersection $\vec{e}$ of the scanner's line-of-sight with the body B.

The approach to determining this function may be outlined as follows. A basic coordinate system is located at the center of the body. The spacecraft position vector $\vec{s}$ brings us to a point in space at which we orient the satellite axes using $\vec{s}$ and the spacecraft velocity vector $\vec{v}$. Relative to the spacecraft axes, yaw, pitch, and roll is introduced, reorienting the spacecraft. Taking into account the pointing direction of the scanner relative to the spacecraft axes, the direction of the scanner line-of-sight can now be determined relative to the basic coordinate system. Where this vector intersects the body B is the sought after result and is measured as the vector $\vec{e}$ relative to the basic coordinate system. See Figure 1. The function at one instant in time would appear as

$$\vec{e} = f(\vec{s}, \vec{v}, y, p, r, \vec{w}, B).$$

Development of this function will appear in a later section.

Description of Coordinate Systems and the Body Surface

The basic coordinate system to which all other coordinate systems and vectors are referenced is a body-centered Cartesian coordinate system. It is oriented so that the X-Y-Z axes coincide with the principle diameters of a spheroid which is used to model the body surface. The X and Y axes coincide with the two principle diameters of equal length while the Z axis will coincide with the third principle diameter. The equation of the body surface is

$$\frac{X^2 + Y^2}{a^2} + \frac{Z^2}{c^2} = 1$$

The value a may be thought of as the equatorial radius of the body while c would be the polar radius. Hence, in order to describe the body B one needs to specify a and c. The satellite coordinate system is related to the basic coordinate system by the position vector s. The satellite coordinate system about which yaw, pitch, and roll are measured is defined as follows: The roll axis is taken to be coincident with the velocity vector $\vec{v}$, the pitch axis is taken to be $\vec{v} \times \vec{s}$ which is normal to the orbital plane and the roll axis, and the yaw axis is taken to be $\vec{v} \times (\vec{v} \times \vec{s})$ and shall be denoted as $\vec{n}$. See Figure 1. This definition of satellite axes orientation is arbitrary. It could equally well have been established by requiring the yaw axis to coincide with $-\vec{s}$, the pitch axis to be $-\vec{s} \times \vec{v}$, and the roll axis to be $(\vec{s} \times \vec{v}) \times \vec{s}$. It seems that different systems are established for different spacecraft. Therefore, if the reader must work with a system other than the one established above, he should make the appropriate substitutions throughout the remainder of the paper.

Measurement of Scanner Orientation

It remains to define a convention for measurement of the scanner's line-of-sight relative to the spacecraft axes. The convention used in this paper is to specify rotations about the yaw, pitch, and roll axes (in that order). The rotations must be such as to rotate a vector initially aligned with the yaw axis $\vec{n}$ so that it will coincide with the scanner's line-of-sight. All rotations follow the standard right hand rule. These three rotational values will be used to form the vector $\vec{w}$ where w_1 is the required rotation about the yaw axis, w_2 the rotation about the pitch axis, and w_3 the rotation about the roll axis.

In the example shown in Figure 2 the line-of-sight vector is described as $\vec{w} = (0°, 0°, -30°)$ since it is in the plane spanned by the pitch and yaw axes; in order to align the yaw axis with the line of sight vector it is necessary only to rotate (in the right hand sense) about the roll axis by $-30°$.



Fig. 1. Description of notation.



FIG. 2. Example of line-of-sight orientation.

POSITIONAL VALUES AS FUNCTIONS OF TIME

Each positional value described up to this point is in reality a function of time. The vector $\vec{s}$ describing the scanner's position is perhaps the most obvious example. Even the vector $\vec{w}$ describing the scanner orientation can be a function of time as in the case of the ERTS scanner which swings back and forth like a pendulum.[1] Since this is the case for all values considered except for a and c, the notation $\vec{s}(t)$, $\vec{w}(t)$, etc. will not be used.

MATHEMATICAL FORMULATION OF THE SCANNER'S INTERSECT POINT

First, the orientation of the spacecraft axes relative to the basic coordinate system will be determined. Let $\boldsymbol{D}$ be the orthogonal matrix whose columns are $\vec{c}_1 = \vec{v} / ||\vec{v}||_2$ (roll), $\vec{c}_2 = (\vec{c}_1 \times \vec{s}) / ||\vec{s}||_2$ (pitch), and $\vec{c}_3 = \vec{c}_1 \times \vec{c}_2$ (yaw).

$$\boldsymbol{D} = [\vec{c}_1, \vec{c}_2, \vec{c}_3]$$

Let

$$\begin{aligned} SY &= \sin(\text{yaw}) & CY &= \cos(\text{yaw}) \\ SP &= \sin(\text{pitch}) & CP &= \cos(\text{pitch}) \\ SR &= \sin(\text{roll}) & CR &= \cos(\text{roll}) \end{aligned}$$

and form M as the product of the three rotation matrices

$$\boldsymbol{M} = \begin{bmatrix} CY & -SY & 0 \\ SY & CY & 0 \\ 0 & 0 & 1 \end{bmatrix} \begin{bmatrix} CP & 0 & SP \\ 0 & 1 & 0 \\ -SP & 0 & CP \end{bmatrix} \begin{bmatrix} 1 & 0 & 0 \\ 0 & CR & -SR \\ 0 & SR & CR \end{bmatrix}$$

$$\boldsymbol{M} = \begin{bmatrix} CY \cdot CP & CY \cdot SP \cdot SR - SY \cdot CR & CY \cdot SP \cdot CR + SY \cdot SR \\ SY \cdot CP & SY \cdot SP \cdot SR + CY \cdot CR & SY \cdot SP \cdot CR - CY \cdot SR \\ -SP & CP \cdot SR & CP \cdot CR \end{bmatrix} \quad (*)$$

Then $\boldsymbol{F} = \boldsymbol{DM}$ is an orthogonal matrix whose column vectors are the roll, pitch, and yaw axes of the spacecraft in the basic coordinate system after yaw, pitch, and roll rotations have been introduced. Next, the orientation of the scanner's line-of-sight relative to the basic coordinate system will be determined. As discussed in the section regarding measurement of scanner orientation, the pointing direction of the scanner is given by specifying rotations about the yaw, pitch, and roll axes in that order. The rotations must be such as to align the spacecraft yaw axis with the scanner line-of-sight. Because these rotations are made in the same fashion as yaw, pitch, and roll attitude corrections, a matrix $\boldsymbol{M}'$ may be formed which is identical to $\boldsymbol{M}$ except that it has the angle w_1 in place of yaw, w_2 in place of pitch, and w_3 in place of roll. Recall that the third column of $\boldsymbol{F}$ represented the yaw axis in the basic coordinate system. Then $\boldsymbol{G} = \boldsymbol{FM}'$ must be an orthogonal matrix whose third column is the unit vector representing the scanner's line-of-sight in the basic coordinate system. Let m be the third column of $\boldsymbol{M}'$ so that

$$\vec{m} = \begin{pmatrix} Cw_1 \cdot Sw_2 \cdot Cw_3 + Sw_1 \cdot Sw_3 \\ Sw_1 \cdot Sw_2 \cdot Cw_3 - Cw_1 \cdot Sw_3 \\ Cw_2 \cdot Cw_3 \end{pmatrix} \qquad (**)$$

Then $\vec{g} = (\boldsymbol{F}\vec{m})$ is the unit vector representing the scanner's line-of-sight in the basic coordinate system.

It remains to determine the intersection of $\vec{g}$ with the surface. This may be done most simply by considering the parametric representations of the surface and the straight line passing through the location of the scanner and aligned with $\vec{g}$. The spheroidal surface can be represented as

$$E(u_1, u_2) = \begin{pmatrix} a \cos u_1 \cos u_2 \\ a \sin u_1 \cos u_2 \\ c \sin u_2 \end{pmatrix}$$

The line is represented as

$$L(u) = \vec{s} + u\,\vec{g}$$

Equating the two

$$\begin{aligned} s_x + u\,g_x &= a \cos(u_1)\cos(u_2) \\ s_y + u\,g_y &= a \sin(u_1)\cos(u_2) \\ s_z + u\,g_z &= c \sin(u_2) \end{aligned}$$

The parameter u may be solved for by multiplying the third equation by a/c, then squaring both sides of each equation, and finally adding all three equations together. This eliminates the parameters u_1 and u_2 and gives a quadratic equation in u.

$$\left[c^2(g_x^2+g_y^2) + a^2 g_z^2\right]u^2 + 2\left[c^2(s_x g_x + s_y g_y) + a^2 s_z g_z\right]u + c^2(s_x^2+s_y^2) + a^2(s_z^2 - c^2) = 0$$

Making the appropriate substitutions, this equation may be represented as

$$Au^2 + 2Bu + C = 0$$

From this quadratic form

$$u = \frac{-B-\sqrt{B^2-AC}}{A}$$

Figure 3 shows the geometric significance of the constant u. If the scanner is to view the Earth, u must be a real number greater than zero. Note that the negative value of the radical is used because when the line intersects the surface in two points we are interested in only the point closest to the scanner. This is the case since the vector $\vec{g}$ is of unit length by its



FIG. 3. Formulation of intersect point.

formulation and therefore in the parametric representation L (u) of the line-of-sight vector, u will be the distance from the scanner to the intersect point.

As can be seen in Figure 3 the intersect point is given as the vector $\boldsymbol{e}$ where

$$\vec{e} = \vec{s} + u\,\vec{g}$$

Summary of Computations

Given: $\vec{s}$ position vector of scanner

$\vec{v}$ velocity vector (roll axis)

$\vec{w}$ orientation of scanner line-of-sight relative to spacecraft axes

p, y, r pitch, yaw, and roll of spacecraft

a equatorial radius of surface

c polar radius of surface

For the sake of coding efficiency the intersect point is found by going through the computations described previously but in the following order:

(1) form the vector $\vec{m}$ as defined by (**)

(2) form the product $\vec{r} = \boldsymbol{M}\vec{m}$ where $\boldsymbol{M}$ is defined by (*)

(3) $\vec{c}_1 = \vec{v} / \|\vec{v}\|_2$

(4) $\vec{c}_2 = (\vec{c}_1 \times \vec{s}) / \|\vec{s}\|_2$

(5) $\vec{c}_3 = \vec{c}_1 \times \vec{c}_2$

let $\boldsymbol{D} = [\vec{c}_1, \vec{c}_2, \vec{c}_3]$

(6) form the vector $\vec{g} = \boldsymbol{D}\vec{r}$

(7) $A = c^2(g^2_x + g^2_y) + a^2 g^2_z$

(8) $B = c^2\,(s_X g_X + s_Y g_Y) + a^2 s_z g_z$

(9) $C = c^2\,(s^2_x + s^2_Y) + a^2\,(s^2_z - c^2)$

(10) $u = \dfrac{-B - \sqrt{B^2 - AC}}{A}$

(11) $\vec{e} = \vec{s} + u\,\vec{g}$

FORTRAN Implementation

Figure 4 is the listing of a FORTRAN implementation of the scheme described in the previous section. The listing presents the implementation as a double precision computation only for the sake of producing high accuracy results. Double precision is not necessary because of any need to overcome numerical conditioning problems. In fact, a user wishing to implement the routine in a production mode might consider using small angle approximations for some of the trigonometric functions as well as reverting to single precision in order to speed up calculations.

Comments

The user should assure himself that the values he has for yaw, pitch, and roll were measured in that order since that is the order of rotation made in this formulation.

The resulting vector $\boldsymbol{e}$ can be converted to geodetic or geocentric latitude and longitude values upon return from the subroutine using the formulae

$$\text{longitude} = \tan^{-1}\left(\frac{e_2}{e_1}\right)$$

$$\text{geocentric latitude} = \tan^{-1}\left(\frac{e_3}{\sqrt{e^2_1 + e^2_2}}\right)$$

$$\text{geodetic latitude} = \tan^{-1}\left(\frac{a^2}{c^2} \times \tan \text{geocentric latitude}\right)$$

```
C
C     SUBROUTINE LOCATE...................................................
C
C FUNCTION...
C     TO LOCATE THE (X,Y,Z) COORDINATES ON THE SURFACE OF
C     A SPHEROID WHERE A SATELLITE BASED SCANNNER IS POINTING.
C INPUT PARAMETERS..........
C     S.......VECTOR OF DIMENSION 3 GIVING THE POSITION OF THE SCANNER
C             RELATIVE TO THE BASIC COORDINATE SYSTEM. (SEE COMMENTS)
C     V.......VECTOR OF DIMENSION 3 GIVING THE VELOCITY OF
C             THE SATELLITE MEASURED RELATIVE TO THE BASIC COORDINATES.
C     W.......VECTOR OF DIMENSION 3 GIVING THE ORIENTAION OF THE
C             SCANNER LINE OF SIGHT RELATIVE TO THE SPACECRAFT AXES.
C             W(1) = ROTATION IN RADIANS ABOUT YAW AXIS
C             W(2) = ROTATION IN RADIANS ABOUT PITCH AXIS
C             W(3) = ROTATION IN RADIANS ABOUT ROLL AXIS
C             NECESSARY TO ALLIGN A VECTOR INITIALLY COINCIDENT WITH
C             THE SPACECRAFT YAW AXIS TO THE LINE OF SIGHT OF THE
C             SCANNER. ROTATIONS MUST BE DETERMINED IN THE ORDER LISTED.
C     Y.......YAW OF SPACECRAFT IN RADIANS
C     P.......PITCH OF SPACECRAFT IN RADIANS
C     R.......ROLL OF SPACECRAFT IN RADIANS
C     EQRAD...EQUATORIAL RADIUS OF BODY (SEE COMMENTS)
C     POLRAD..POLAR RADIUS OF BODY (SEE COMMENTS)
C OUTPUT PARAMETERS..........
C     E.......SOLUTION VECTOR OF DIMENSION 3 RETURNING (X,Y,Z)
C             CO-ORDINATES OF THE INTERSECT OF THE LINE OF SIGHT
C             WITH THE BODY SURFACE. MEASURED IN BASIC SYSTEM.
C     MISS....RETURN FLAG. MISS = 0 IMPLIES SCANNER SEES BODY.
C             MISS = 1 IMPLIES SCANNER DOES NOT SEE THE BODY.
C             MISS = 2 IMPLIES SCANNER LOOKS DIRECTLY AWAY FROM BODY.
C COMMENTS...
C     THE BASIC COORDINATE SYSTEM IS DEFINED AS A CARTESIAN COORDINATE
C     SYSTEM AT WHICH THE SPHEROIDAL SURFACE IS CENTERED. THE
C     SPHEROIDAL SURFACE IS DESCRIBED BY
C             X*X+Y*Y         Z*Z
C           ----------- + ------------- = 1
C           EQRAD*EQRAD   POLRAD*POLRAD
C PRECISION...  DOUBLE
C REQ'D ROUTINES... FORTRAN TRIG AND SQUARE ROOT FUNCTIONS
C AUTHOR... E. PUCCINELLI   GSFC  CODE 931  8-1-75
C
      SUBROUTINE LOCATE(S,V,W,P,Y,R,EQRAD,POLRAD,E,MISS)
      IMPLICIT REAL*8(A-H,O-Z)
      DIMENSION S(3),V(3),W(3),E(3)
      MISS= 0
      ER2 = EQRAD*EQRAD
      PR2 = POLRAD*POLRAD
C..........SET TRIG VALUES..........
      CY  = DCOS(Y)
      CP  = DCOS(P)
      CR  = DCOS(R)
      SY  = DSIN(Y)
      SP  = DSIN(P)
      SR  = DSIN(R)
      CSY = DCOS(W(1))
      CSP = DCOS(W(2))
      CSR = DCOS(W(3))
      SNY = DSIN(W(1))
      SNP = DSIN(W(2))
      SNR = DSIN(W(3))
C..........FORM VECTOR M..........
      EM1 = CSY*SNP*CSR + SNY*SNR
      EM2 = SNY*SNP*CSR - CSY*SNR
      EM3 = CSP*CSR
C..........FORM PRODUCT OF MATRIX CAP.M * VECTOR M
      R1  = EM1*CY*CP + EM2*(CY*SP*SR - SY*CR) + EM3*(CY*SP*CR + SY*SR)
      R2  = EM1*SY*CP + EM2*(SY*SP*SR + CY*CR) + EM3*(SY*SP*CR - CY*SR)
      R3  = -EM1*SP + EM2*CP*SR + EM3*CP*CR
C..........FORM MATRIX D
      VLG = DSQRT(V(1)*V(1) + V(2)*V(2) + V(3)*V(3))
      D11 = V(1)/VLG
      D21 = V(2)/VLG
      D31 = V(3)/VLG
      SLG = DSQRT(S(1)*S(1) + S(2)*S(2) + S(3)*S(3))
      D12 = (D21*S(3) - S(2)*D31)/SLG
      D22 = (D31*S(1) - S(3)*D11)/SLG
      D32 = (D11*S(2) - S(1)*D21)/SLG
      D13 = D21*D32 - D22*D31
      D23 = D31*D12 - D32*D11
      D33 = D11*D22 - D12*D21
C..........FORM UNIT VECTOR G  =  D * M * (VECTOR M)
      GX  = R1*D11 + R2*D12 + R3*D13
      GY  = R1*D21 + R2*D22 + R3*D23
      GZ  = R1*D31 + R2*D32 + R3*D33
C..........FORM U3 ... DISTANCE OF SCANNER TO INTERCEPT POINT......
      A   = PR2*(GX*GX + GY*GY) + ER2*GZ*GZ
      B   = PR2*(S(1)*GX + S(2)*GY) + ER2*S(3)*GZ
      C   = PR2*(S(1)*S(1) + S(2)*S(2)) + ER2*(S(3)*S(3) - PR2)
      F   = B*B - A*C
C..........IF F IS NEGATIVE, SCANNER DOES NOT SEE BODY..........
      IF(F.GE.0.D0) GO TO 10
      MISS= 1
      RETURN
   10 U3  = -(B+DSQRT(F))/A
C..........IF U3 IS NEGATIVE SCANNER IS LOOKING DIRECTLY AWAY FROM BODY
      IF(U3.GT.0.D0) GO TO 20
      MISS= 2
      RETURN
C..........FORM ANSWER VECTOR E ... THE INTERCEPT POINT..........
   20 E(1)= S(1) + U3*GX
      E(2)= S(2) + U3*GY
      E(3)= S(3) + U3*GZ
      RETURN
      END
```

FIG. 4. FORTRAN listing.

This conversion was not included in the subroutine in order to keep the coding as brief as possible.

ATTITUDE DETERMINATION

There is a growing tendency to digitally rectify satellite scanner images by first using several geographically located points on an unrectified image to compute a more accurate attitude model of the satellite than is provided by the satellite instrumentation.[2,3] The idea is to take a known model (such as described in this paper)

$$\vec{e} = f(\vec{s}, \vec{v}, \vec{w}, p, y, r, a, c)$$

and, for several different values of $\vec{e}$, determine by least squares or digital filtering techniques, functions which approximate $\vec{s}$, p, y, r over some period of time.

ACKNOWLEDGMENTS

The author gratefully acknowledges the comments made by Messrs. R. Desai and H. Siddalingaiah of Computer Sciences Corporation in discussions on this paper.

REFERENCES

1. *ERTS Data User's Handbook*, Doc 71SD4249, NASA
2. Caron, R. H. and Simon, K. W., "Rectification of Spaceborne Imagery", *Journal of Spacecraft*, Vol. 12, No. 1., Jan. 1975, pp. 27-32.
3. Bernstein, R., *All Digital Processing of ERTS Images*, final report by IBM for Goddard Space Flight Center, contract NAS5-21716, April 1975.
4. Kreyzig, *Differential Geometry*, University of Toronto Press, Toronto, 1959.